

## Revisão e introdução ao Portugol Studio – Exemplos e Exercícios

### Importante

Sítio oficial

- <http://lite.acad.univali.br/portugol/>

Canal no YouTube

- <https://www.youtube.com/watch?v=K02TnB3IGnQ&list=PLb9yvNDCid3jQAEbNoPHtPROSWwmRSM-t>

### 1. [Exemplo 01] – Entrada e saída de valores

#### VisuAlg

```
algoritmo "Exemplo 02"

var
idade : inteiro
nome: caracter
inicio
    escreva("Informe seu nome: ")
    leia(nome)
    escreva("Informe sua idade: ")
    leia(idade)
    escreval("Olá ",nome," , você tem ",idade," anos.")
finalgoritmo
```

#### Portugol Studio

```
programa
{
    funcao inicio ()
    {
        cadeia nome
        inteiro idade

        escreva("Informe seu nome: ")
        leia(nome)
        escreva("Informe sua idade: ")
        leia(idade)
        escreva("Olá ",nome," , você tem ",idade," anos.")
    }
}
```

Observação:

## 2. [Exemplo 02] – Operações aritméticas

### VisuAlg

```
algoritmo "Exemplo 02"

var
numA, numB, resultado : real
inicio
    escreva("Informe o primeiro número: ")
    leia(numA)
    escreva("Informe o segundo número: ")
    leia(numB)
    resultado <- numA + numB
    escreval("A soma dos números é: ", resultado)
    resultado <- numA - numB
    escreval("A subtração dos números é: ", resultado)
    resultado <- numA / numB
    escreval("A divisão dos números é: ", resultado)
    resultado <- numA * numB
    escreval("A multiplicação dos números é: ", resultado)
finalgoritmo
```

### Portugol Studio

```
programa
{
    funcao inicio()
    {
        real numA, numB, resultado

        escreva("Informe o primeiro número: ")
        leia(numA)
        escreva("Informe o segundo número: ")
        leia(numB)
        resultado = numA + numB
        escreva("A soma dos números é: ", resultado, "\n")
        resultado = numA - numB
        escreva("A subtração dos números é: ", resultado, "\n")
        resultado = numA / numB
        escreva("A divisão dos números é: ", resultado, "\n")
        resultado = numA * numB
        escreva("A multiplicação dos números é: ", resultado, "\n")
    }
}
```

Observação:

### 3. [Exemplo 03] – Estrutura condicional simples e composta

#### VisuAlg

```
algoritmo "Exemplo 03"
var
idade : inteiro
nome: caracter
inicio
escreva("Informe seu nome: ")
leia(nome)
escreva("Informe sua idade: ")
leia(idade)
se (idade < 0) entao
    escreval("Idade invalida!")
senao
    se (idade = 0) entao
        escreval("Olá ",nome, ", você nasceu a pouco tempo")
    senao
        se (idade > 0) e (idade < 18) entao
            escreval("Olá ",nome, ", você é menor de idade")
        senao
            se (idade >= 18) entao
                escreval("Olá ",nome, ", você é maior de idade")
            fimse
        fimse
    fimse
fimse
finalgoritmo
```

Observação:

#### Portugol Studio

```
programa
{
    funcao inicio ()
    {
        cadeia nome
        inteiro idade

        escreva("Informe seu nome: ")
        leia(nome)
        escreva("Informe sua idade: ")
        leia(idade)

        se (idade < 0)
        {
            escreva("Idade invalida!")
        }
        senao
        {
            se (idade == 0)
            {
                escreva("Olá ",nome, ", você nasceu a pouco tempo")
            }
            senao
            {
                se (idade > 0 e idade < 18)
                {
                    escreva("Olá ",nome, ", você é menor de idade")
                }
                senao
                {
                    se (idade >= 18)
                    {
                        escreva("Olá ",nome, ", você é maior de idade")
                    }
                }
            }
        }
    }
}
```

#### 4. [Exemplo 04] – Estruturas de Repetição (Enquanto)

##### VisuAlg

```
algoritmo "Exemplo 04"

var
idade : inteiro
nome: caracter
inicio
idade <- -1
escreva("Informe seu nome: ")
leia(nome)
enquanto (idade < 0) faca
    escreva("Informe sua idade: ")
    leia(idade)
    se (idade < 0) entao
        escreval("Idade inválida!")
    fimse
fimenquanto
escreval("Olá ",nome,",", você tem ",idade," anos.")
finalgoritmo
```

Observação:

##### Portugol Studio

```
programa
{
    funcao inicio()
    {
        cadeia nome
        inteiro idade

        idade = -1

        escreva("Informe seu nome: ")
        leia(nome)

        enquanto (idade < 0)
        {
            escreva("Informe sua idade: ")
            leia(idade)
            se (idade < 0)
            {
                escreva("Idade inválida! \n")
            }
        }

        escreva("Olá ",nome,",", você tem ",idade," anos.")
    }
}
```

## 5. [Exemplo 05] – Estruturas de Repetição (Para)

### VisuAlg

```
algoritmo "Exemplo 05"

var
quantidade, idade, total: inteiro
media: real
inicio
para quantidade de 1 ate 10 passo 1 faca
    escreva("Informa a idade da ",quantidade,"° pessoa: ")
    leia(idade)
    total <- total + idade
fimpara
media <- total / 10
escreval("A média geral (aritmética) das idades é: ", media)
fimalgoritmo
```

Observação:

### Portugol Studio

```
programa
{
    funcao inicio()
    {
        inteiro quantidade, idade, total
        real media

        total = 0

        para (quantidade = 1; quantidade <= 10; quantidade++)
        {
            escreva("Informa a idade da ",quantidade,"° pessoa: ")
            leia(idade)
            total = total + idade
        }

        media = total / 10

        escreva("A média geral (aritmética) das idades é: ", media)
    }
}
```

## 6. [Exemplo 05] – Estruturas de Repetição (Repita)

### VisuAlg

```
algoritmo "Exemplo 06"

var
numero, maior: inteiro
resposta: caracter

inicio

maior <- 0

repita
    escreva("Informe um número: ")
    leia(numero)
    se (numero > maior) entao
        maior <- numero
    fimse
    escreva("Deseja continuar (sim ou não)? ")
    leia(resposta)

ate (resposta = "não")

escreval("O maior número informado é: ", maior)

fimalgoritmo
```

Observação:

### Portugol Studio

```
programa
{
    funcao inicio()
    {
        inteiro numero, maior
        cadeia resposta

        maior = 0

        faca
        {
            escreva("Informe um número: ")
            leia(numero)
            se (numero > maior)
            {
                maior = numero
            }
            escreva("Deseja continuar (sim ou não)? ")
            leia(resposta)
        }
        enquanto (resposta == "sim")

        escreva("O maior número informado é: ", maior)
    }
}
```

## 7. [Exemplo 07] – Estrutura de dados homogênea unidimensional – Vetor

### VisuAlg

```
algoritmo "Exemplo 07"

var
posicao, valor: inteiro
numeros: vetor[1..5] de inteiro

inicio

posicao <- 1

enquanto (posicao <= 5) faca
    escreva("Informe um número: ")
    leia(valor)
    se ( (valor mod 2) = 0 ) entao
        numeros[posicao] <- valor
        posicao <- posicao + 1
        escreval("O número foi armazenado no vetor")
    senao
        escreval("O número NÃO foi armazenado no vetor")
    fimse
fimenquanto

escreval("Números armazenados no vetor:")

para posicao de 1 ate 5 passo 1 faca
    escreval(numeros[posicao])
fimpara

finalgoritmo
```

Observação:

### Portugol Studio

```
programa
{
    funcao inicio()
    {
        inteiro posicao, valor
        inteiro numeros[5]

        posicao = 0

        enquanto (posicao <= 4)
        {
            escreva("Informe um número: ")
            leia(valor)
            se (valor % 2 == 0)
            {
                numeros[posicao] = valor
                posicao = posicao + 1
                escreva("O número foi armazenado no vetor \n")
            }
            senao
            {
                escreva("O número NÃO foi armazenado no vetor \n")
            }
        }

        escreva("Números armazenados no vetor: \n")

        para(posicao = 0; posicao <=4; posicao++)
        {
            escreva(numeros[posicao], "\n")
        }
    }
}
```

8. **[Exercício 1]** - Reescreva o algoritmo abaixo no Portugol Studio.

```
algoritmo "prova"
var
numeros: vetor[1..10] de inteiro
numero, indiceA, indiceB: inteiro
inicio
para indiceB de 1 ate 10 passo 1 faca
    escreva("Informe um número:")
    leia(numeros[indiceB])
fimpara
para indiceA de 1 ate 9 passo 1 faca
    para indiceB de (indiceA + 1) ate 10 passo 1 faca
        se (numeros[indiceA] < numeros[indiceB]) entao
            numero <- numeros[indiceB]
            numeros[indiceB] <- numeros[indiceA]
            numeros[indiceA] <- numero
        fimse
    fimpara
fimpara
para indiceB de 1 ate 10 faca
    se ((numeros[indiceB] mod 2) = 0) entao
        escreva (numeros[indiceB])
    fimse
fimpara
fimalgoritmo
```

9. **[Exercício 2]** - Escrever um algoritmo (Portugol Studio) para determinar o consumo médio de um automóvel sendo fornecida a distância total percorrida pelo automóvel e o total de combustível gasto.

10. **[Exercício 3]** - Faça um algoritmo (Portugol Studio) para calcular e informar a área do Círculo.

- Fórmula para o cálculo:  $area = PI * r^2$
- $PI = 3,14$   $r =$  raio

11. **[Exercício 4]** - Crie um algoritmo (Portugol Studio) que simule uma disputa de PAR ou ÍMPAR entre duas pessoas. O algoritmo deve solicitar os nomes dos jogadores e suas opções (1: PAR ou 2: ÍMPAR). O algoritmo deve gerar um número aleatório e definir o vencedor da partida. O vencedor da disputa será o jogador que vencer três partidas, ou seja, melhor de cinco.

12. **[Exercício 5]** - Faça um algoritmo (Portugol Studio) que calcule e informe o valor da S:

$$S = 1 + 3/2 + 5/3 + 7/4 + \dots + 99/50$$

13. **[Exercício 6]** - Elabore um algoritmo (Portugol Studio) que solicite ao usuário o tipo e valor de sua aplicação. O algoritmo deve calcular e armazenar os valores da aplicação corrigidos por um ano. Por fim, o algoritmo deve os valores corrigidos por mês.

Tipos de aplicação financeira:

- 1 – Poupança: taxa de juros 10%
- 2 - Letra de Crédito Imobiliário: taxa de juros 15%
- 3 - Letra de Crédito do Agronegócio: taxa de juros 18%

Por exemplo:

```
Informe o valor da aplicação: 100
1 - Poupança (10%)
2 - Letra de Crédito Imobiliário (15%)
3 - Letra de Crédito do Agronegócio (18%)
Informe o tipo da aplicação: 1
Valores corrigidos por mês
1º: 110.00
2º: 121.00
3º: 133.10
4º: 146.41
5º: 161.05
6º: 177.16
7º: 194.87
8º: 214.36
9º: 235.79
10º: 259.37
11º: 285.31
12º: 313.84
```